
Change 1

	Testcase name
	11.2

	Reason for change
	According to 34.229 (section 11.2.4), new pairs of security association should be set up in step 5 after the 401 Unauthorized message is received and the second Register message is sent over them. 

For this reason new function is included to only generate new RAND and set new SPIs acaccording to Register message in step 3. Theese values are then passed to 401 Unauthorized mesasge according to specs in step 4. 

New set of SAs is then set up after the 401 Unauthorized mesasge is sent. After this, the next Register message is sent over new pair of SAs in step 5.
Originally the SS sent the 401 Unauthorized message to UE in step 3  over new set of SAs but the UE did not have the the SAs established and therefore could not receive the 401 Unauthorized message.

	Summary of change
	Added new function f_IMS_Generate_New_RAND_and_SPI  in the beginning of step 4 and moved f_IMS_Register_SecurityInit to the end of the step 4.

	Source of change
	IMS\11\IMS_CC_NotificationTestcases.ttcn


Before change:

	function f_TC_11_2_IMS() runs on IMS_PTC

  {

    var SecurityClientParams_Type v_SecurityClientParams;

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var REGISTER_Request v_RegisterReq;

    var SUBSCRIBE_Request v_SubscribeReq;

    var charstring v_NW_AddrStr;

    var SipUrl v_ContactUrl;

    var template (value) Reginfo_Type v_XmlMessage;

    var template (value) RegInfo_RegistrationList_Type v_Registrations;

    var charstring v_PublicUserIdentity;

    var boolean v_ApplyGRUU := false;

    var integer v_NonceCount;

    var boolean v_IsReregistration;

    var integer v_ExpirationTime := 60;

    var float v_ReRegistrationTime := int2float(v_ExpirationTime / 2);

    timer t_TRegExpireMin := f_IMS_SetTimerToleranceMin(v_ReRegistrationTime);

    timer t_TRegExpireMax := f_IMS_SetTimerToleranceMax(v_ReRegistrationTime);

    f_IMS_CC_Preamble(IPCAN_SignallingOnly, IMS_NULL, IMS_Security);

    f_IMS_CC_StartSignalling(IPCAN_InitialRegistration);

    v_RegisterReq := f_IMS_Register_NonGIBA();

    v_SubscribeReq := f_IMS_Register_SubscribeNotify(v_RegisterReq);

    v_ContactUrl := f_MessageHeader_GetContactSipUrl(v_RegisterReq.msgHeader);

    v_PublicUserIdentity := f_IMS_PTC_ImsInfo_GetURI(firstId);           // @sic R5s130266 change 1.1 sic@

    f_IMS_TestBody_Set(true);

    v_NW_AddrStr := f_IMS_PTC_NW_Address_GetAddrStr();    /* @sic R5s130133 change 6: f_IMS_PTC_NW_Address_GetHostFormat replaced by f_IMS_PTC_NW_Address_GetAddrStr sic@ */

    // Set the 60 seconds expiration time in the XML body

    v_Registrations[0] := f_XML_RegistrationInfo(v_RegisterReq, v_PublicUserIdentity, "a100", "980", active, shortened, v_ExpirationTime, v_ApplyGRUU);

    v_XmlMessage := f_XML_RegInfo(partial, 1, v_Registrations);

    f_IMS_Register_Notify(v_ContactUrl, v_SubscribeReq, v_XmlMessage, 2);

    f_IMS_CC_ReleaseConnection(IPCAN_InitialRegistration);

    // Step 3

    t_TRegExpireMin.start;  // @sic R5s130266 change 1.2 - MCC160 Implementation sic@

    t_TRegExpireMax.start;  // @sic R5s130266 change 1.2 - MCC160 Implementation sic@

    f_IMS_CC_StartSignalling(IPCAN_MO_IMS_Signalling);

    v_NonceCount := 2;

    v_IMS_DATA_REQ := f_IMS_PTC_ReRegistration_WaitForRegistration(t_TRegExpireMin, t_TRegExpireMax, v_NonceCount);

    v_RegisterReq := v_IMS_DATA_REQ.Request.Register;

    v_SecurityClientParams := f_IMS_PTC_ReRegistration_SecurityClientParamsGet();

    f_IMS_PTC_ReRegistration_SecurityClientParamsCheck(v_RegisterReq, v_SecurityClientParams);

    // Step 4

    f_IMS_Register_SecurityInit(v_RegisterReq, not4b tsc_IMS_AuthRAND);    /* generate a new RAND */

    IMS_Server.send(cas_IMS_DATA_RSP(f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo), cs_Response(c_statusLine401, f_IMS_RegisterResponse_401_MessageHeaderTX(v_RegisterReq))));

    // Step 5

    v_NonceCount := 1;             // nonce counter is reset to 1 due to re-authentication @sic R5s130266 change 2 - MCC160 Implementation sic@

    v_IsReregistration := true;    // @sic R5s130266 change 2 - MCC160 Implementation sic@

    v_IMS_DATA_REQ := f_IMS_REGISTER_SubsequentRequest(-, -, v_NonceCount, v_IsReregistration);    // @sic R5s130266 change 2 - MCC160 Implementation sic@

    v_RegisterReq := v_IMS_DATA_REQ.Request.Register;

    // Step 6

    IMS_Server.send(cas_IMS_DATA_RSP(f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo), cs_Response(c_statusLine200, f_IMS_RegisterResponse_200_MessageHeaderTX(v_RegisterReq))));

    f_IMS_PreliminaryPass(__FILE__, __LINE__, "test case 11.2");

    f_IMS_TestBody_Set(false);

    f_IMS_CC_Postamble(IPCAN_MO_IMS_Signalling);

  }

}


After change :

	function f_TC_11_2_IMS() runs on IMS_PTC

  {

    var SecurityClientParams_Type v_SecurityClientParams;

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var REGISTER_Request v_RegisterReq;

    var SUBSCRIBE_Request v_SubscribeReq;

    var charstring v_NW_AddrStr;

    var SipUrl v_ContactUrl;

    var template (value) Reginfo_Type v_XmlMessage;

    var template (value) RegInfo_RegistrationList_Type v_Registrations;

    var charstring v_PublicUserIdentity;

    var boolean v_ApplyGRUU := false;

    var integer v_NonceCount;

    var boolean v_IsReregistration;

    var integer v_ExpirationTime := 60;

    var float v_ReRegistrationTime := int2float(v_ExpirationTime / 2);

    timer t_TRegExpireMin := f_IMS_SetTimerToleranceMin(v_ReRegistrationTime);

    timer t_TRegExpireMax := f_IMS_SetTimerToleranceMax(v_ReRegistrationTime);

    f_IMS_CC_Preamble(IPCAN_SignallingOnly, IMS_NULL, IMS_Security);

    f_IMS_CC_StartSignalling(IPCAN_InitialRegistration);

    v_RegisterReq := f_IMS_Register_NonGIBA();

    v_SubscribeReq := f_IMS_Register_SubscribeNotify(v_RegisterReq);

    v_ContactUrl := f_MessageHeader_GetContactSipUrl(v_RegisterReq.msgHeader);

    v_PublicUserIdentity := f_IMS_PTC_ImsInfo_GetURI(firstId);           // @sic R5s130266 change 1.1 sic@

    f_IMS_TestBody_Set(true);

    v_NW_AddrStr := f_IMS_PTC_NW_Address_GetAddrStr();    /* @sic R5s130133 change 6: f_IMS_PTC_NW_Address_GetHostFormat replaced by f_IMS_PTC_NW_Address_GetAddrStr sic@ */

    // Set the 60 seconds expiration time in the XML body

    v_Registrations[0] := f_XML_RegistrationInfo(v_RegisterReq, v_PublicUserIdentity, "a100", "980", active, shortened, v_ExpirationTime, v_ApplyGRUU);

    v_XmlMessage := f_XML_RegInfo(partial, 1, v_Registrations);

    f_IMS_Register_Notify(v_ContactUrl, v_SubscribeReq, v_XmlMessage, 2);

    f_IMS_CC_ReleaseConnection(IPCAN_InitialRegistration);

    // Step 3

    t_TRegExpireMin.start;  // @sic R5s130266 change 1.2 - MCC160 Implementation sic@

    t_TRegExpireMax.start;  // @sic R5s130266 change 1.2 - MCC160 Implementation sic@

    f_IMS_CC_StartSignalling(IPCAN_MO_IMS_Signalling);

    v_NonceCount := 2;

    v_IMS_DATA_REQ := f_IMS_PTC_ReRegistration_WaitForRegistration(t_TRegExpireMin, t_TRegExpireMax, v_NonceCount);

    v_RegisterReq := v_IMS_DATA_REQ.Request.Register;

    v_SecurityClientParams := f_IMS_PTC_ReRegistration_SecurityClientParamsGet();

    f_IMS_PTC_ReRegistration_SecurityClientParamsCheck(v_RegisterReq, v_SecurityClientParams);

    // Step 4

    f_IMS_Generate_New_RAND_and_SPI(v_RegisterReq, not4b tsc_IMS_AuthRAND);
    IMS_Server.send(cas_IMS_DATA_RSP(f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo), cs_Response(c_statusLine401, f_IMS_RegisterResponse_401_MessageHeaderTX(v_RegisterReq))));

    f_IMS_Register_SecurityInit(v_RegisterReq, not4b tsc_IMS_AuthRAND,-,true);
    // Step 5

    v_NonceCount := 1;             // nonce counter is reset to 1 due to re-authentication @sic R5s130266 change 2 - MCC160 Implementation sic@

    v_IsReregistration := true;    // @sic R5s130266 change 2 - MCC160 Implementation sic@

    v_IMS_DATA_REQ := f_IMS_REGISTER_SubsequentRequest(-, -, v_NonceCount, v_IsReregistration);    // @sic R5s130266 change 2 - MCC160 Implementation sic@

    v_RegisterReq := v_IMS_DATA_REQ.Request.Register;

    // Step 6

    IMS_Server.send(cas_IMS_DATA_RSP(f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo), cs_Response(c_statusLine200, f_IMS_RegisterResponse_200_MessageHeaderTX(v_RegisterReq))));

    f_IMS_PreliminaryPass(__FILE__, __LINE__, "test case 11.2");

    f_IMS_TestBody_Set(false);

    f_IMS_CC_Postamble(IPCAN_MO_IMS_Signalling);

  }

}


Change 2
	Testcase name
	11.2

	Reason for change
	Function f_IMS_PTC_Security_Init for initialising or re-initialising SPIs and protected ports on the network side is always executing and causes problems during re-authentication procedure as this function was already called in step 4 and the second execution causes changes of ports to old ones. For this reason new input parameter p_IsRe_auth is included to be used in condition checker to avoid function f_IMS_PTC_Security_Init in case of re-authentication.

	Summary of change
	Added input parameter p_IsRe_auth for function function f_IMS_Register_SecurityInit and condition checker for function f_IMS_PTC_Security_Init

	Source of change
	Common\IMS\IMS_Procedures_Registration.ttcn


Before change:

	function f_IMS_Register_SecurityInit(REGISTER_Request p_RegisterReq,

                                       B128_Type p_Rand := tsc_IMS_AuthRAND,

                                       boolean p_Unprotected := false) runs on IMS_PTC

  { /* @sic R5s120858 change 6..8 - MCC160 implementation: additional parameter p_Unprotected to deal with GIBA sic@ */

    /* @sic R5s130266 change 1.3, 4, 5 - MCC160 Implementation: SPIs for re-authentication sic@ */

    var Common_AuthenticationParams_Type v_AuthenticationParams := f_IMS_AuthenticationInit(p_Rand);

    var IP_AddrInfo_Type v_UE_Address := f_IMS_PTC_UE_Address_Get();

    var IP_AddrInfo_Type v_NW_Address := f_IMS_PTC_NW_Address_Get();

    var SipUrl v_ContactUrl := f_MessageHeader_GetContactSipUrl(p_RegisterReq.msgHeader);

    var PortNumber_Type v_Port_us;

    var PortNumber_Type v_Port_uc;

    var IPsec_SPI_Type v_SPI_us;

    var IPsec_SPI_Type v_SPI_uc;

    if (p_Unprotected) {

      f_IMS_PortsAndSecurityConfig(f_SIP_ContactUrl_GetPort(v_ContactUrl),         // may be omit in which case 5060 is chosen by the IP PTC

                                   v_NW_Address,

                                   v_UE_Address);

    }

    else {

      f_IMS_InstallIPsecKeys(v_AuthenticationParams.IK, v_AuthenticationParams.CK);

      v_Port_us := str2int(f_SIP_Register_GetSecurityParam(p_RegisterReq, "port-s"));

      v_Port_uc := str2int(f_SIP_Register_GetSecurityParam(p_RegisterReq, "port-c"));

      v_SPI_us  := str2int(f_SIP_Register_GetSecurityParam(p_RegisterReq, "spi-s"));

      v_SPI_uc  := str2int(f_SIP_Register_GetSecurityParam(p_RegisterReq, "spi-c"));

      /* initialise or re-initialise SPIs and protected ports of the network side: */

      f_IMS_PTC_Security_Init(v_AuthenticationParams, v_Port_us, v_Port_uc, v_SPI_us, v_SPI_uc);

      /* configure IMS security; any existing security context will be released

       * NOTE: release of the security context is not done at the end of the deregistration since it is not sure that the deregistration really happens and

       *       when realeasing the security directly after sending the deregistration it may lead to timing issues */

      f_IMS_PortsAndSecurityConfig(f_SIP_ContactUrl_GetPort(v_ContactUrl),         // may be omit in which case 5060 is chosen by the IP PTC

                                   v_NW_Address,

                                   v_UE_Address,

                                   cs_IMS_SecurityInfo(f_IMS_PTC_Security_GetProtectedPorts(),

                                                       f_IMS_PTC_Security_GetSPIs(),

                                                       f_IMS_PTC_Security_GetIntegrityAlgorithm(),

                                                       f_IMS_PTC_Security_GetCipheringAlgorithm()));

    }

  }


After change :

	function f_IMS_Register_SecurityInit(REGISTER_Request p_RegisterReq,

                                       B128_Type p_Rand := tsc_IMS_AuthRAND,

                                       boolean p_Unprotected := false,

                                       boolean p_IsRe_auth := false) runs on IMS_PTC 

  { /* @sic R5s120858 change 6..8 - MCC160 implementation: additional parameter p_Unprotected to deal with GIBA sic@ */

    /* @sic R5s130266 change 1.3, 4, 5 - MCC160 Implementation: SPIs for re-authentication sic@ */

    var Common_AuthenticationParams_Type v_AuthenticationParams := f_IMS_AuthenticationInit(p_Rand);

    var IP_AddrInfo_Type v_UE_Address := f_IMS_PTC_UE_Address_Get();

    var IP_AddrInfo_Type v_NW_Address := f_IMS_PTC_NW_Address_Get();

    var SipUrl v_ContactUrl := f_MessageHeader_GetContactSipUrl(p_RegisterReq.msgHeader);

    var PortNumber_Type v_Port_us;

    var PortNumber_Type v_Port_uc;

    var IPsec_SPI_Type v_SPI_us;

    var IPsec_SPI_Type v_SPI_uc;

    if (p_Unprotected) {

      f_IMS_PortsAndSecurityConfig(f_SIP_ContactUrl_GetPort(v_ContactUrl),         // may be omit in which case 5060 is chosen by the IP PTC

                                   v_NW_Address,

                                   v_UE_Address);

    }

    else {

      f_IMS_InstallIPsecKeys(v_AuthenticationParams.IK, v_AuthenticationParams.CK);

      v_Port_us := str2int(f_SIP_Register_GetSecurityParam(p_RegisterReq, "port-s"));

      v_Port_uc := str2int(f_SIP_Register_GetSecurityParam(p_RegisterReq, "port-c"));

      v_SPI_us  := str2int(f_SIP_Register_GetSecurityParam(p_RegisterReq, "spi-s"));

      v_SPI_uc  := str2int(f_SIP_Register_GetSecurityParam(p_RegisterReq, "spi-c"));

      /* initialise or re-initialise SPIs and protected ports of the network side: */

      if (not p_IsRe_auth) { 

//this part is avoided after re-authentication -> p_IsRe_auth==TRUE


        f_IMS_PTC_Security_Init(v_AuthenticationParams, v_Port_us, v_Port_uc, v_SPI_us, v_SPI_uc); //already initializated in f_IMS_Generate_New_RAND_and_SPI

      }











      /* configure IMS security; any existing security context will be released

       * NOTE: release of the security context is not done at the end of the deregistration since it is not sure that the deregistration really happens and

       *       when realeasing the security directly after sending the deregistration it may lead to timing issues */

      f_IMS_PortsAndSecurityConfig(f_SIP_ContactUrl_GetPort(v_ContactUrl),         // may be omit in which case 5060 is chosen by the IP PTC

                                   v_NW_Address,

                                   v_UE_Address,

                                   cs_IMS_SecurityInfo(f_IMS_PTC_Security_GetProtectedPorts(),

                                                       f_IMS_PTC_Security_GetSPIs(),

                                                       f_IMS_PTC_Security_GetIntegrityAlgorithm(),

                                                       f_IMS_PTC_Security_GetCipheringAlgorithm()));

    }

  }


Change 3
	Testcase name
	11.2

	Reason for change
	New function that only generates new RAND and sets new SPI ports on network side, which are then used in 401 Unauthorized message sent from SS to UE as response to REGISTER message in step 3.

	Summary of change
	Added new function to generate RAND and to set the SPIs used in 401 Unauthorized message in step 4. 

	Source of change
	Common\IMS\IMS_Procedures_Registration.ttcn


New function:

	function f_IMS_Generate_New_RAND_and_SPI(REGISTER_Request p_RegisterReq,







  B128_Type p_Rand := tsc_IMS_AuthRAND,

 







  boolean p_Unprotected := false) runs on IMS_PTC

  { 

    var Common_AuthenticationParams_Type v_AuthenticationParams := f_IMS_AuthenticationInit(p_Rand);

    var PortNumber_Type v_Port_us;

    var PortNumber_Type v_Port_uc;

    var IPsec_SPI_Type v_SPI_us;

    var IPsec_SPI_Type v_SPI_uc;

    if (not p_Unprotected)

    {

      v_Port_us := str2int(f_SIP_Register_GetSecurityParam(p_RegisterReq, "port-s"));

      v_Port_uc := str2int(f_SIP_Register_GetSecurityParam(p_RegisterReq, "port-c"));

      v_SPI_us  := str2int(f_SIP_Register_GetSecurityParam(p_RegisterReq, "spi-s"));

      v_SPI_uc  := str2int(f_SIP_Register_GetSecurityParam(p_RegisterReq, "spi-c"));

      /* initialise or re-initialise SPIs and protected ports of the network side: */

      f_IMS_PTC_Security_Init(v_AuthenticationParams, v_Port_us, v_Port_uc, v_SPI_us, v_SPI_uc);

    }

  }


Change 4
	Testcase name
	11.2

	Reason for change
	TcpConnections.Protected_pc_us and Protected_uc_ps are set to omit only loacally and are not passed to other IMS handlers updated. For this reason inout parameter needs to be included for p_ImsServer to update the changes made to IMS server. 

	Summary of change
	Added inout key word to input parameters of function fl_ImsServer_SecurityStop to update p_ImsServer.

	Source of change
	Common\IP_PTC\IP_PTC_IMS_Handler.ttcn


Before change:

	function fl_ImsServer_SecurityStop(IMS_Server_Type p_ImsServer) runs on IP_PTC

  {

    var IP_AddrInfo_Type v_NW_Address;

    var IP_AddrInfo_Type v_UE_Address;

    var IMS_SecurityInfo_Type v_SecurityInfo;

    var PortNumber_Type v_LocalPortServer;

    var PortNumber_Type v_LocalPortClient;

    if (fl_ImsServer_SecurityIsEstablished(p_ImsServer)) {

      v_NW_Address := p_ImsServer.RegistrationInfo.NW_Address;

      v_UE_Address := p_ImsServer.RegistrationInfo.UE_Address;

      v_SecurityInfo := p_ImsServer.RegistrationInfo.SecurityInfo;

      v_LocalPortServer := v_SecurityInfo.ProtectedPorts.Port_ps;

      v_LocalPortClient := v_SecurityInfo.ProtectedPorts.Port_pc;

      // stop UDP client and server

      f_UdpClient_Stop(v_NW_Address, v_LocalPortClient);

      f_UdpServer_Stop(v_NW_Address, v_LocalPortServer);

      // stop TCP server

      f_TcpClient_CloseConnection(p_ImsServer.TcpConnections.Protected_pc_us);

      f_TcpServer_CloseConnection(p_ImsServer.TcpConnections.Protected_uc_ps);

      p_ImsServer.TcpConnections.Protected_pc_us := omit;

      p_ImsServer.TcpConnections.Protected_uc_ps := omit;

      f_TcpServer_Stop(v_NW_Address, v_LocalPortServer);

      // Release security association

      fl_IPsec_Release(v_NW_Address, v_UE_Address, v_SecurityInfo);

    }

  }


After change :

	function fl_ImsServer_SecurityStop(inout IMS_Server_Type p_ImsServer) runs on IP_PTC
  {

    var IP_AddrInfo_Type v_NW_Address;

    var IP_AddrInfo_Type v_UE_Address;

    var IMS_SecurityInfo_Type v_SecurityInfo;

    var PortNumber_Type v_LocalPortServer;

    var PortNumber_Type v_LocalPortClient;

    if (fl_ImsServer_SecurityIsEstablished(p_ImsServer)) {

      v_NW_Address := p_ImsServer.RegistrationInfo.NW_Address;

      v_UE_Address := p_ImsServer.RegistrationInfo.UE_Address;

      v_SecurityInfo := p_ImsServer.RegistrationInfo.SecurityInfo;

      v_LocalPortServer := v_SecurityInfo.ProtectedPorts.Port_ps;

      v_LocalPortClient := v_SecurityInfo.ProtectedPorts.Port_pc;

      // stop UDP client and server

      f_UdpClient_Stop(v_NW_Address, v_LocalPortClient);

      f_UdpServer_Stop(v_NW_Address, v_LocalPortServer);

      // stop TCP server

      f_TcpClient_CloseConnection(p_ImsServer.TcpConnections.Protected_pc_us);

      f_TcpServer_CloseConnection(p_ImsServer.TcpConnections.Protected_uc_ps);

      p_ImsServer.TcpConnections.Protected_pc_us := omit;

      p_ImsServer.TcpConnections.Protected_uc_ps := omit;

      f_TcpServer_Stop(v_NW_Address, v_LocalPortServer);

      // Release security association

      fl_IPsec_Release(v_NW_Address, v_UE_Address, v_SecurityInfo);

    }

  }

